
TECHNICAL SUPPORT • MAY 2000

BY JOHN E.  JOHNSTON
Enterprise Networking

A Windows Scripting
Command Line Utility Example

I n the March 2000 issue of Technical
Support, I wrote an introductory column

on Windows Scripting. Since its publica-
tion, several readers have emailed me, asking
for a command line Windows Scripting
example. This type of utility script can be
used to automate a variety of functions such
as re-booting file servers, starting a backup
job or monitoring disk space usage. One of
the most common requests I received was
for a sample utility that would copy files
from one file server to another. This month,
I will review a sample Visual Basic script
that copies all of the files from a directory
residing on a NetWare server to a backup
directory residing on a Windows NT server.
The functions performed in the script are
as follows:

● maps network drives to the Novell and
Windows NT server

● performs a DOS xcopy command to
copy the data

● removes the network drive mappings
created previously

The entire script is illustrated in Figure 1.
Let's take a closer look at each function per-
formed in this script.

Mapping Network Drives
The first section of the script creates the

Network object, which is required when
issuing any type of network functions
within a script. The first two lines of the
script create this Network object. The first
line of the script — Dim WSHNetwork —
is simply declaring a variable named
WSHNetwork. The second line calls the
Wscript. CreateObject function to create
the Network object. A pointer to the
Network object is then placed in our
WSHNetwork variable. We will use this

pointer to perform our drive mappings and
drive mapping deletes.

The next few lines of code create the
Drive1 and Share1 variables and then
equate a value to these variables. Notice
that the Share1 variable is set to
\\nw51\sys. This is the NetWare file server
and the NetWare volume name that contain

the data we are planning to copy to the
Windows NT server.

The next line calls the MapNetworkDrive
function, passing the Drive1 and Share1
parameters. The MapNetworkDrive func-
tion in this section of code performs the
same function as the following command to
map a network drive:

' Command Line Utility Script 
'

' Create the Network Object
'
Dim WSHNetwork
Set WSHNetwork = WScript.CreateObject("WScript.Network")

' Map a drive for the NetWare server
'

Dim Drive1
Dim Share1
Drive1 = "J:"
Share1 = "\\nw51\sys"
WSHNetwork.MapNetworkDrive Drive1, Share1

'Map a drive for the Windows NT server
'

Dim Drive2
Dim Share2
Drive2 = "K:"
Share2 = "\\tsg1_pdc\backup"
WSHNetwork.MapNetworkDrive Drive2, Share2

'Create the Shell object and execute the xcopy command
'

Dim WSHShell
set WSHShell = Wscript.CreateObject("Wscript.Shell")
iRetCode = WSHShell.Run("command.com /k xcopy j:\data\*.* k:", 1, TRUE)

' Delete the Network drive mappings created earlier
'

WSHNetwork.RemoveNetworkDrive Drive1
WSHNetwork.RemoveNetworkDrive Drive2

' Clean up and exit
'

Wscript.Quit

FIGURE 1: SAMPLE COMMAND LINE UTILITY SCRIPT TO COPY
FILES FROM A NETWARE SERVER TO A WINDOWS NT SERVER

net use J: \\nw51\sys

Note: The MapNetworkDrive function,
when utilized for a NetWare drive mapping,
can only map the drive to the NetWare vol-
ume level. In other words, we can't map the
NetWare drive to the directory level.

The next few lines of code map the K:
drive to a Windows NT Server share named
\\tsg1_pdc\backup.

Executing a DOS command
Now that we have our drives mapped for

the NetWare and the Windows NT servers,
we are ready to issue the DOS xcopy com-
mand to copy the data from the NetWare
server to the Windows NT server. The
first two lines of this section of code
define the WSHShell variable, then the
Wscript.CreateObject is called to create the
Wscript.Shell object. This object is required
for issuing DOS commands within a script.

Next, we issue the Run function of the
Shell to perform the xcopy command.
Notice that we are calling command.com to
issue the xcopy command. Let's take a look
at each parameter passed to the Run function:

● command.com — This calls the
Windows command processor.

● /k — This causes the DOS window to
be left open after the script executes.
The user must exit from the DOS
window before the script will end. This
will allow the user to check the output
of the xcopy command. If you do not
want this DOS box to remain open, use
the /c parameter instead.

● xcopy j:\data\*.* k: — This is the actual
xcopy command. It will copy all of the
files from the data directory to the
share named BACKUP on the Windows
NT server.

● "1" — This is the WindowStyle property.
A "1" causes the DOS window to be dis-
played on the user's screen. A "2" causes
the window to start minimized. A "3"
causes the window to start maximized.

● TRUE — This parameter tells the script
to wait for the command to complete
before continuing with the script. A
value of FALSE would cause the script
to continue with the next line of script
code right after issuing the xcopy com-

mand. This would cause trouble in our
script because the next lines of the script
would delete the network drive mappings
before the xcopy had completed.

Deleting the Network Mappings
The last section of code calls the

RemoveNetworkDrive function to delete
the drive mappings created previously. To
execute this sample script, you can either
utilize any of the following methods:

● cscript.exe interface (i.e., C:>cscript.exe
c:\path and file name of script file)

● double click on the script file from the
Explorer application

● create a shortcut to the script file on a
user's desktop

or, you could also use a scheduling package,
such as WINAT, to automatically call this
script at a certain time every day. This
sample script was written in a very simple
manner to show you the power of
Windows scripting.

Note: This sample has absolutely no error
checking built into it.

CONCLUSION

If this column has peaked your interest in
Windows scripting, you may want to practice
by taking this sample script and modifying

it a bit. One of the sample scripts that comes
with the Microsoft Script Host download
(www.microsoft.com/vbscript/scripting.asp)
contains a section of code that automates
the process of mapping network drives. This
code segment, illustrated in Figure 2,
attempts to map the Z: drive, and if unsuc-
cessful, it attempts to map the Y: drive, and
so on until an unused drive letter is found.
Integrating this type of logic in our sample
script would make it much more robust, as
it would prevent the need for the hard coded
drive letters.

Happy scripting! If you have any questions
or comments on this material, or have sug-
gestions for future topics, please feel free to
email me at johnj@fast.net.  

NaSPA member John E. Johnston is the owner
of a web development company that specializes
in e-commerce. John also performs contract
work on Novell, NT and Unix networks. He can
be contacted at johnj@fast.net.

Function TryMapDrive(intDrive, strShare)
Dim strDrive
strDrive = Chr(intDrive + 64) & ":"
On Error Resume Next
WSHNetwork.MapNetworkDrive strDrive, strShare
TryMapDrive = Err.Number = 0

End Function

For intDrive = 26 To 5 Step -1
If TryMapDrive(intDrive, strShare) Then Exit For

Next

If intDrive <= 5 Then
MsgBox "Unable to connect to network share. " & _

"There are currently no drive letters available for use. " & _
CRLF & _
"Please disconnect one of your existing network connections " & _
"and try this script again. ", _
vbExclamation + vbOkOnly,        _
L_Welcome_MsgBox_Title_Text

Else
strDrive = Chr(intDrive + 64) & ":"
MsgBox "Connected " & strShare & " to drive " & strDrive,   _
vbInformation + vbOkOnly,                          _

L_Welcome_MsgBox_Title_Text

FIGURE 2: SAMPLE CODE SEGMENT TO AUTOMATE THE DRIVE MAPPING PROCESS 


